Lesson 31: Recurrences, iteration and approximation

```
> restart;
  read "d:/m210/fibonacci.txt";
```

Some Fibonacci puzzles

These are some problems from the Fibonacci Quarterly that can be solved with the help of **mpow** and **fib4**.

(1) Let H_n be any solution (in integers) of the recurrence $H_n = H_{n-1} + H_{n-2}$. Show that $7\,H_n \equiv H_{n+15} \textbf{ mod } 10$.

Solution: If $X(n) = \langle H(n-1), H(n)\rangle$, then $X(n+1) = M \,.\, X(n)$, so $X(n+15) = M^{15} \,.\, X(n)$

```
> M . <H(n-1), H(n)>;
```

$$\begin{bmatrix} H(n) \\ H(n-1) + H(n) \end{bmatrix} \qquad (1.1)$$

```
> H(n+15) = (M^15 . <H(n-1),H(n)>)[2];
```

$$H(n+15) = 610\,H(n-1) + 987\,H(n)$$

```
> % mod 10;
```

$$H(n+15) = 7\,H(n)$$

(2). Show that F_{5^n} is divisible by 5^n (but not by 5^{n+1}).

You get from one power of 5 to the next by multiplying by 5. Can **fib4** tell us a useful identity?

```
> fib4(5*k);
```

$$\left(\left(F_{k-1}^2 + F_k^2\right)\left(F_{k-1} F_k + F_k\left(F_k + F_{k-1}\right)\right) + \left(F_{k-1} F_k + F_k\left(F_k + F_{k-1}\right)\right)\left(F_k^2 + \left(F_k + F_{k-1}\right)^2\right)\right) F_{k-1} + \left(\left(F_{k-1} F_k + F_k\left(F_k + F_{k-1}\right)\right)^2 + \left(F_k^2 + \left(F_k + F_{k-1}\right)^2\right)^2\right) F_k$$

```
> factor(%);
```

$$5\,F_k\left(F_{k-1}^4 + 2\,F_{k-1}^3 F_k + 4\,F_{k-1}^2 F_k^2 + 3\,F_k^3 F_{k-1} + F_k^4\right)$$

So F_{5k} is divisible by $5\,F_k$, i.e. every time we multiply k by 5 we get an additional factor of 5. Since $F_{5^0} = F_1 = 1$ is divisible by 5^0, we have by mathematical induction F_{5^n} is divisible by 5^n for every nonnegative n.

Could F_{5^n} be divisible by a higher power of 5? Certainly not for $n = 0$ or $n = 1$.

If it's true for the first time for $n = m + 1$, then with $k = 5^m$ we must have

$F_{k-1}^4 + 2\,F_{k-1}^3 F_k + 4\,F_{k-1}^2 F_k^2 + 3\,F_k^3 F_{k-1} + F_k^4$

divisible by 5. But F_k is already divisible by 5, so F_{k-1}^4 would have to be divisible by 5. And then F_{k-1} would be divisible by 5. But if F_k and F_{k-1} were divisible by 5, then all the Fibonacci numbers would be divisible by 5, and F_1 certainly isn't.

(3) What is a_n, if $a_{n+1} = 5\,a_n^3 - 3\,a_n$ for nonnegative integers n with $a_0 = 1$?

This is another recurrence relation, but a nonlinear one. Let's start by calculating a few terms

```
> a[0]:= 1;
  for nn from 0 to 5 do
    a[nn+1]:= 5*a[nn]^3 - 3*a[nn]
  end do;
```

$$a_0 := 1$$

$$a_1 := 2$$

$$a_2 := 34$$

$$a_3 := 196418$$

$$a_4 := 37889062373143906$$

$$a_5 := 271964099255182923543922814194423915162591622175362$$

$a_6 :=$

1005784040477634373949250581829128589677588166452408546692608465300231\
4710064957975890680217086513480428319808835573562763479871653927751530\
4438631163554

Maybe it's good to have stopped here: the numbers are growing very rapidly. This being a problem that's supposed to have something to do with Fibonacci numbers, we might remember that $a_2 = 34$ is a Fibonacci number

```
> fib4(9);
```

$$34$$

Of course $a_1 = 2$ is also a Fibonacci number, namely F_3. You might make a wild guess from this, that $a_n = F_{3^n}$. Let's test it for these first few:

```
> seq(a[n]-fib4(3^n),n=1..6);
```

$$0, 0, 0, 0, 0, 0$$

It seems to work. Now can we prove it?

```
> fib4(3*k);
```

$$\left(F_{k-1} F_k + F_k \left(F_k + F_{k-1}\right)\right) F_{k-1} + \left(F_k^2 + \left(F_k + F_{k-1}\right)^2\right) F_k$$

This should be $5\,F_k^3 - 3\,F_k$, at least if k is a power of 3, in order for our conjecture to be true.

```
> expand(%);
```

$$3\,F_{k-1}^2 F_k + 3\,F_{k-1} F_k^2 + 2\,F_k^3 \qquad \textbf{(1.2)}$$

What's the difference between that and $5\,F_k^3 - 3\,F_k$?

```
> % - (5*F[k]^3 - 3*F[k]);
```

$$3\,F_{k-1}^2 F_k + 3\,F_{k-1} F_k^2 - 3\,F_k^3 + 3\,F_k$$

```
> factor(%);
```

$$-3\,F_k \left(-F_{k-1}^2 - F_{k-1} F_k + F_k^2 - 1\right)$$

Well, the $-3\,F_k$ isn't 0, so maybe the other factor is.

```
> Q:= k -> -fib4(k-1)^2-fib4(k-1)*fib4(k)+fib4(k)^2-1;
```

$$Q := k \to -fib4(k-1)^2 - fib4(k-1)\,fib4(k) + fib4(k)^2 - 1$$

```
> seq(Q(k),k=1..10);
```

$$0,\ -2,\ 0,\ -2,\ 0,\ -2,\ 0,\ -2,\ 0,\ -2$$

It looks like this should be 0 for odd k and -2 for even k. That will be fine for our purposes: the powers of 3 are all odd. Can we prove it by mathematical induction?

If the pattern works, Q(k+1) should be -2 when Q(k) = 0 and 0 when Q(k) = -2,
so in any case Q(k) + Q(k+1) = -2.

```
> Q(k) + Q(k+1);
```

$$-F_{k-1}^2 - F_{k-1}F_k - 2 - F_k\left(F_k + F_{k-1}\right) + \left(F_k + F_{k-1}\right)^2$$

```
> normal(%);
```

$$-2$$

Yes, it works. Working backwards, we have a proof:
This calculation showed that $Q(k) + Q(k+1) = -2$.
We know $Q(1) = 0$.
Then by mathematical induction we get that $Q(k) = 0$ if k is odd and -2 if k is even.
Maple showed that $F_{3k} - \left(5\,F_k^3 - 3\,F_k\right) = -3\,F_k\,Q(k)$, so if k is odd (and in particular if k is a power of 3) this is 0.
That says that $a_n = F_{3^n}$ satisfies the recurrence relation $a_{n+1} = 5\,a_n^3 - 3\,a_n$. Since it also satisfies the initial condition $a_0 = 1$, by mathematical induction this must be a_n.

By the way, for fans of linear algebra, there's another way to look at the equation

$$Q(k) = \begin{cases} 0 & k \text{ is odd} \\ -2 & otherwise \end{cases}$$

```
> -Q(k) - 1 = (-1)^k;
```

$$F_{k-1}^2 + F_{k-1}F_k - F_k^2 = (-1)^k$$

The left side is the determinant of M^k:

```
> mpow(k);
```

$$\begin{bmatrix} F_{k-1} & F_k \\ F_k & F_k + F_{k-1} \end{bmatrix}$$

```
> LinearAlgebra[Determinant](%);
```

$$F_{k-1}^2 + F_{k-1}F_k - F_k^2$$

But $det\left(M^k\right) = det(M)^k$, and

```
> LinearAlgebra[Determinant](M);
```

$$-1$$

Solving recurrences

Maple has a command for solving recurrence relations, i.e. getting an explicit formula for their solutions: **rsolve**. Here's the first recurrence relation we looked at in Lesson 29:

```
> a:= 'a':
  rsolve({a(0)=1, a(n)=2*a(n-1)}, a(n));
```

$$2^n$$

It works for some nonlinear recurrence relations, but not many. Here's the one from the last Fibonacci puzzle:

```
> rsolve({a(n+1) = 5*a(n)^3-3*a(n), a(0)=1},a(n));
```

$$rsolve(\{a(0)=1, a(n+1)=5\,a(n)^3-3\,a(n)\}, a(n))$$

Let's try an easier one.

```
> rsolve({a(n)=3*a(n-1)^2,a(0)=2},a(n));
```

$$\frac{1}{3}\,2^{2^n}\,3^{2^n} \tag{2.1}$$

But it's pretty good for the linear ones. How about the Fibonacci relation?

```
> sol := rsolve({f(n)=f(n-1)+f(n-2),f(0)=0,f(1)=1},f(n));
```

$$sol := \frac{1}{5}\sqrt{5}\left(\frac{1}{2}+\frac{1}{2}\sqrt{5}\right)^n - \frac{1}{5}\sqrt{5}\left(-\frac{1}{2}\sqrt{5}+\frac{1}{2}\right)^n$$

Well, if it works that's another way to calculate the Fibonacci numbers.

```
> fib5:= unapply(%,n);
```

$$fib5 := n \rightarrow \frac{1}{5}\sqrt{5}\left(\frac{1}{2}+\frac{1}{2}\sqrt{5}\right)^n - \frac{1}{5}\sqrt{5}\left(-\frac{1}{2}\sqrt{5}+\frac{1}{2}\right)^n$$

```
> fib5(4);
```

$$\frac{1}{5}\sqrt{5}\left(\frac{1}{2}+\frac{1}{2}\sqrt{5}\right)^4 - \frac{1}{5}\sqrt{5}\left(-\frac{1}{2}\sqrt{5}+\frac{1}{2}\right)^4$$

```
> expand(%);
```

$$3$$

I might put the **expand** into the function.

```
> fib6:= n -> expand(fib5(n));
```

$$fib6 := n \rightarrow expand(fib5(n))$$

```
> seq(fib4(n)=fib6(n),n=0..10);
```

$$0=0,\ 1=1,\ 1=1,\ 2=2,\ 3=3,\ 5=5,\ 8=8,\ 13=13,\ 21=21,\ 34=34,\ 55=55$$

OK, it seems to work. Can we prove it? To prove it by mathematical induction, knowing that **fib6(0)** and **fib6(1)** are F_0 and F_1 respectively, we just have to prove that it satisfies the recurrence relation.

```
> fib6(n)-(fib6(n-1)+fib6(n-2));
```

$$\frac{1}{5}\sqrt{5}\left(\frac{1}{2}+\frac{1}{2}\sqrt{5}\right)^n - \frac{1}{5}\sqrt{5}\left(-\frac{1}{2}\sqrt{5}+\frac{1}{2}\right)^n - \frac{1}{5}\,\frac{\sqrt{5}\left(\frac{1}{2}+\frac{1}{2}\sqrt{5}\right)^n}{\frac{1}{2}+\frac{1}{2}\sqrt{5}}$$

$$+\frac{1}{5}\frac{\sqrt{5}\left(-\frac{1}{2}\sqrt{5}+\frac{1}{2}\right)^n}{-\frac{1}{2}\sqrt{5}+\frac{1}{2}}-\frac{1}{5}\frac{\sqrt{5}\left(\frac{1}{2}+\frac{1}{2}\sqrt{5}\right)^n}{\left(\frac{1}{2}+\frac{1}{2}\sqrt{5}\right)^2}$$

$$+\frac{1}{5}\frac{\sqrt{5}\left(-\frac{1}{2}\sqrt{5}+\frac{1}{2}\right)^n}{\left(-\frac{1}{2}\sqrt{5}+\frac{1}{2}\right)^2}$$

```
> normal(%);
```

$$0$$

Binet's formula, the Golden Ratio and Fibonacci

For those inclined to linear algebra, $\frac{1}{2}+\frac{\sqrt{5}}{2}$ and $\frac{1}{2}-\frac{\sqrt{5}}{2}$ are the eigenvalues of the matrix M, and this formula can be obtained by diagonalizing M.

```
> LinearAlgebra[Eigenvalues](M);
```

$$\begin{bmatrix}\frac{1}{2}+\frac{1}{2}\sqrt{5}\\ -\frac{1}{2}\sqrt{5}+\frac{1}{2}\end{bmatrix}$$

The formula is called "Binet's formula", although Binet wasn't the first person to find it.

$\frac{1}{2}+\frac{\sqrt{5}}{2}$ is a famous number: it's sometimes called the **Golden Ratio** and denoted by the Greek letter ϕ. *The DaVinci Code* contains all sorts of nonsense about it (to be fair, such nonsense appears in lots of other places too).

```
> phi:= 1/2 + sqrt(5)/2; evalf(phi);
```

$$\phi := \frac{1}{2}+\frac{1}{2}\sqrt{5}$$

$$1.618033988$$

To Golden Ratio enthusiasts, just about any ratio that's close to 1.6 (e.g. the ratio between your height and the height of your navel) is this number.
See e.g. <http://www.maa.org/devlin/devlin_06_04.html>.

ϕ and $\frac{1}{2}-\frac{\sqrt{5}}{2}$ (which is $-\frac{1}{\phi}$) are the roots of the polynomial x^2-x-1.

```
> solve(x^2-x-1, x);
```

$$\frac{1}{2}+\frac{1}{2}\sqrt{5},\ -\frac{1}{2}\sqrt{5}+\frac{1}{2}$$

That polynomial enters into this because it is the characteristic polynomial of the matrix *M*.

```
> LinearAlgebra[CharacteristicPolynomial](M,x);
```

$$x^2-x-1 \qquad \textbf{(3.1)}$$

Binet's formula is not necessarily a good way to calculate particular values of F_n.
If you want to do an exact calculation, you must expand out the n'th powers (as we did in **fib6**).
Each will involve $n+1$ terms, of which the even numbered ones cancel and the odd ones are the same for both ϕ^n and $\left(-\frac{1}{\phi}\right)^n$, leaving about $\frac{n}{2}$ rational numbers to add.
It's easier to use repeated squaring on the matrix, where the number of steps is essentially proportional to $\log(n)$.
On the other hand, you could use **evalf**, in which case the answer will only be approximate when n is large, unless you use a high value for **Digits**.

```
> evalf(fib5(100));
```

$$3.542248316\ 10^{20}$$

```
> Digits:= 22: evalf(fib5(100));
```

$$3.542248481792619150622\ 10^{20}$$

```
> round(%);
```

$$354224848179261915062$$

```
> fib4(100);
```

$$354224848179261915075$$

Not quite enough digits...

```
> Digits:= 25: evalf(fib5(100));
```

$$3.542248481792619150750036\ 10^{20}$$

```
> round(%);
```

$$354224848179261915075$$

What the formula is really good for is giving the asymptotic behaviour of F_n for large n.

Since $\left|-\frac{1}{\phi}\right| < 1 < |\phi|$, F_n is approximately $\frac{\phi^n}{\sqrt{5}}$, with $\left|F_n - \frac{\phi^n}{\sqrt{5}}\right| = \left|\frac{\left(-\frac{1}{\phi}\right)^n}{\sqrt{5}}\right| < \frac{1}{2}$ for all $n \geq 0$ and converging to 0 as $n \to \infty$.

```
> fibapp:= n -> phi^n/sqrt(5);
```

$$fibapp := n \to \frac{\phi^n}{\sqrt{5}}$$

For even n, F_n is slightly less than **fibapp(n)**, for odd n it's slightly greater.

```
> Digits:= 20:
  fib4(30), evalf(fibapp(30));
```

$$832040,\ 8.3204000000024037296\ 10^{5} \tag{3.2}$$

```
> fib4(31), evalf(fibapp(31));
```

$$1346269,\ 1.3462689999998514411\ 10^{6} \tag{3.3}$$

```
> (fibapp(n)=37889062373143906);
```

$$\frac{1}{5}\sqrt{5}\left(\frac{1}{2}+\frac{1}{2}\sqrt{5}\right)^n = 37889062373143906 \tag{3.4}$$

```
> invfib1:= x -> log[phi](x*sqrt(5));
```

$$invfib1 := x \to \log_{\phi}\left(x\sqrt{5}\right) \quad (3.5)$$

```
> evalf(invfib1(37889062373143906));
```

$$81.000000000000000001 \quad (3.6)$$

```
> fib4(81);
```

$$37889062373143906 \quad (3.7)$$

A numerical iteration

Recurrence relations are often used in numerical methods. You want to compute some sequence y_n which satisfies a recurrence relation, so you start with known values for y_0 or the first few y_n, and iterate the recurrence formula. This may or may not be a good way to calculate y_n. The main thing that can go wrong is that the inevitable roundoff errors can grow with each iteration until they overwhelm the true solution.

For example, suppose you want to approximate the following sequence of numbers, related to the remainders in the series for *e*:

$w_n = n!\left(\mathrm{e} - \left(\sum_{k=0}^{n} \frac{1}{k!}\right)\right)$. How could you compute them?

Maple commands introduced in this lesson:

rsolve